
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc92457117]Corrections required for function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8
0.1 [bookmark: _Toc92457118]Correction to f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8
	Function name
	f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8()

	Reason for change
	In the function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8, at Step 7, PDCP counts are being set based on an activation time. However, at Step 8, whilst re-establishing security, PDCP counts are being obtained with immediate activation time which would not include the increment of the DL PDCP SN for the transmission of the RRCReconfiguration message at Step 2. Hence, this mechanism does not guarantee the values that are obtained at Step 8 are the ones that were set at Step 7.

	Summary of change
	It is recommended to obtain the PDCP counts independent of the flag p_SetGetOfPDCPCount and increment the DL PDCP SN for the transmission of the RRCReconfiguration message at Step 2 and in the case of RBConfig_KeyChange being TRUE, reset the PDCP count to 0 as well as use the obtained PDCP counts at Step 1 to pass as a parameter to functions that re-establish security.

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	Accepted in principle but the proposed implementation will not work with KeyChange and AM DRBs. See MCC160 proposed implementation below.



Before Change:
	function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_NR_CellId,
                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,
                                                       boolean                                p_RBConfig_KeyChange := true,
                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,
                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,
                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,
                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                       template (omit) MeasConfig             p_MeasConfig := omit,
                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,
                                                       boolean p_SetGetOfPDCPCount := true
                                                       ) runs on NR_BASE_PTC
  { //Added p_SetGetOfPDCPCount to be set to false in PDCP tc with TM mode


    var SubFrameTiming_Type v_Timing;
    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_NR_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    
    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).
    // Not needed in PDCP tc with PDCP configured in TM mode
    if (p_SetGetOfPDCPCount){
        v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_NR_CellId, p_NR_RadioBearerList, p_RBConfig_KeyChange);
        // @sic R5s191200 R5s201150 sic@
    }

    /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.
    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */
    if (f_TimingInfo_IsNow(p_TimingInfo)) {
      v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);
    } else {
      v_Timing := valueof(p_TimingInfo.SubFrame);
    }
    
    //Step 2 At T: Send RRCReconfiguration.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));
    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration
       => RLC ACK is not taken into consideration */

    <<SKIPPED CODE>>

    // Step 8 At T + 10ms: Re-establish security, disable TA transmission.
   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.
   if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s191200 R5s200309 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_NR_CellId, p_NCC, cs_TimingInfo_NR(v_Timing)); //@sic R5s200754 sic@
   } else {
      v_RRC_SecurityParams := f_NR_Security_Get();
      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false);
      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering,
                                            cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
   }
  
   // Disable TA transmission
   f_NR_ULGrantConfiguration_Common(p_NR_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);

}




After Change:
	function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_NR_CellId,
                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,
                                                       boolean                                p_RBConfig_KeyChange := true,
                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,
                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,
                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,
                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                       template (omit) MeasConfig             p_MeasConfig := omit,
                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,
                                                       boolean p_SetGetOfPDCPCount := true
                                                       ) runs on NR_BASE_PTC
  { //Added p_SetGetOfPDCPCount to be set to false in PDCP tc with TM mode

   var SubFrameTiming_Type v_Timing;
    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_NR_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    var integer i; //WA#
    
    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).
    // Not needed in PDCP tc with PDCP configured in TM mode
    //WA#
    v_NR_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_NR_CellId);
    v_NR_PdcpCountInfoList[0].DL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList[0].DL.Value) + 1), 32);

    /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.
    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */
    if (f_TimingInfo_IsNow(p_TimingInfo)) {
      v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);
    } else {
      v_Timing := valueof(p_TimingInfo.SubFrame);
    }
    
    //Step 2 At T: Send RRCReconfiguration.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));
    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration
       => RLC ACK is not taken into consideration */

    <<SKIPPED CODE>>

    // Step 8 At T + 10ms: Re-establish security, disable TA transmission.
   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.
   if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s191200 R5s200309 sic@
       //WA#
       for (i := 0; i < lengthof(v_NR_PdcpCountInfoList); i:= i + 1) {
          v_NR_PdcpCountInfoList[i].DL.Value := int2bit(0, 32);
          v_NR_PdcpCountInfoList[i].UL.Value := int2bit(0, 32);
      }
      f_NR_SS_AS_ActivateSecurity_Current(p_NR_CellId, p_NCC, cs_TimingInfo_NR(v_Timing), -,v_NR_PdcpCountInfoList); //@sic R5s200754 sic@ WA#
   } else {
      v_RRC_SecurityParams := f_NR_Security_Get();
      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false,-, v_NR_PdcpCountInfoList); //WA#
      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering,
                                            cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
   }
  
   // Disable TA transmission
   f_NR_ULGrantConfiguration_Common(p_NR_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);

  }



MCC160 proposed implementation
      function f_NR_GetPdcpCountInfoListIntraNR_HO (NR_CellId_Type                              p_SourceCellId,
                                                 template( value) NR_RadioBearerList_Type    p_RadioBearerList,
                                                 boolean                                     p_RBConfig_KeyChange := true,
                                                 boolean                                     p_IncrementSRB1 := true  //@sic R5-216260 sic@
                                                 )  runs on NR_BASE_PTC return NR_PdcpCountInfoList_Type
   { //@sic R5s220123 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList:=f_NR_SS_PdcpCount_Get(p_SourceCellId);
    var integer i; 
    
    // Acc to 38.300 cl 9.2.3.1:
    // For DRBs using RLC AM mode, PDCP can either be re-established together with a security key change or initiate a data recovery procedure without a key change.
    // For DRBs using RLC UM mode and for SRBs, PDCP can either be re-established together with a security key change or remain as it is without a key change
    if (p_RBConfig_KeyChange) {
        //No need to retrieve PDCP count for UM and SRBs because they are re-established in the target cell - PDCP SN reset to 0
        //Retrieve PDCP count for AM DRBs only
        for (i := 0; i < lengthof(v_NR_PdcpCountInfoList); i:= i + 1) {
            if (not(fl_CheckAM_Drb(v_NR_PdcpCountInfoList[i], p_RadioBearerList ))) {
                v_NR_PdcpCountInfoList[i].DL.Value := int2bit(0,32);
                v_NR_PdcpCountInfoList[i].UL.Value := int2bit(0,32);
            }
        }
    } else {
        for (i := 0; i < lengthof(v_NR_PdcpCountInfoList); i:= i + 1) {
            if (not(fl_CheckAM_Drb(v_NR_PdcpCountInfoList[i], p_RadioBearerList)) and not(fl_CheckSrb(v_NR_PdcpCountInfoList[i])) ) {
                v_NR_PdcpCountInfoList[i].DL.Value := int2bit(0,32);
                v_NR_PdcpCountInfoList[i].UL.Value := int2bit(0,32);
            }
        }
        if (p_IncrementSRB1) {//@sic R5-216260 sic@
            // DL PDCP count value for SRB1 shall be incremented by 1 because RRCReconfiguration will be sent out on SRB1 of source Cell.
            //@sic R5s201150 sic@
            v_NR_PdcpCountInfoList[0].DL.Value := int2bit((bit2int(v_NR_PdcpCountInfoList[0].DL.Value) + 1), 32);
        }
    }
    return v_NR_PdcpCountInfoList;
   }  




function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8(NR_CellId_Type                         p_NR_CellId,
                                                       template (value) DL_DCCH_Message       p_RRCReconfiguration,
                                                       boolean                                p_RBConfig_KeyChange := true,
                                                       template (value) NR_RadioBearerList_Type   p_NR_RadioBearerList,
                                                       template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,
                                                       template (value) TimingInfo_Type       p_TimingInfo := cs_TimingInfo_Now,
                                                       RNTI_Value                             p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                       template (omit) MeasConfig             p_MeasConfig := omit,
                                                       NextHopChainingCount                   p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                       template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,
                                                       boolean p_SetGetOfPDCPCount := true
                                                       ) runs on NR_BASE_PTC
  { //Added p_SetGetOfPDCPCount to be set to false in PDCP tc with TM mode
    var SubFrameTiming_Type v_Timing;
    var IntegerList_Type v_DrbIdList := f_NR_GetActiveDRBs(p_NR_CellId);  // List of DRBs to release in the cell

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    
    // Step 1. Before T: Get PDCP count for AM DRBs and SRBs (if applied).
    // Not needed in PDCP tc with PDCP configured in TM mode
    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_NR_CellId, p_NR_RadioBearerList, p_RBConfig_KeyChange);   // @sic R5s191200 R5s201150 R5s220123 sic@

   /* For intra cell HO the HO procedure and the SS configuration must be executed with timing info.
    * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell. */
    if (f_TimingInfo_IsNow(p_TimingInfo)) {
      v_Timing := f_NR_GetNextSendOccasion_DL(p_NR_CellId, tsc_NR_DelayBeforeIntraCellHO);
    } else {
      v_Timing := valueof(p_TimingInfo.SubFrame);
    }
    
    //Step 2 At T: Send RRCReconfiguration.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration));
    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration
       => RLC ACK is not taken into consideration */
    
    //Step 3 At T + 5ms: Release SRBs and DRBs.
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Release(p_NR_CellId, cs_TimingInfo_NR(v_Timing), v_DrbIdList, tsc_NoCnfReq, tsc_FollowOnFlag);   //@sic R5-206369 sic@

    // Step 4 At T + 5ms: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE: Since the RACH procedure may require a new C-RNTI to be used it cannot be configured before sending out the RRCReconfiguration.
    // NOTE 2: The FollowOnFlag is set to true in the ASPs reconfiguring the RadioBearerList (step 3) and C-RNTI (step 4).
    f_NR_SS_C_RNTI_Config(p_NR_CellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing), tsc_NoCnfReq, tsc_FollowOnFlag); //@sic R5s190747 change 5.3 R5-206369 sic@
    f_NR_SS_RachProcedureConfigHO(p_NR_CellId, p_RachProcedureConfig, cs_TimingInfo_NR(v_Timing), p_RACH_ConfigDedicated);
    f_NR_CellInfo_SetRNTI(p_NR_CellId, p_RNTI_Value);
          
    // Step 5 At T + 5ms: (Re-)configure measurement gap if provided in the RRCReconfiguration message.
    // @sic R5-197222 sic@
    if (isvalue(p_MeasConfig.measGapConfig)) {
        f_NR_SS_MeasGapCtrlConfig(p_NR_CellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig), cs_TimingInfo_NR(v_Timing));
    }

    // Step 6 At T + 10ms: (Re-) configure SRBs and DRBs.
    v_Timing := f_SubFrameTiming_AddMilliSeconds(v_Timing, 5);   // add 5ms
    f_NR_SS_SRBs_DRBs_Config(p_NR_CellId, cs_TimingInfo_NR(v_Timing), p_NR_RadioBearerList);

    // Step 7 At T + 10ms:  Restore the PDCP count for AM DRBs and SRBs (if applied).
    //NOTE 2:   For AM DRBs the PDCP count is maintained while for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the  RRCReconfiguration message content
    // Not needed in PDCP tc with PDCP configured in TM mode
    if (p_SetGetOfPDCPCount){
        f_NR_SS_PdcpCount_Set(p_NR_CellId, v_NR_PdcpCountInfoList, cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
    }
   // Step 8 At T + 10ms: Re-establish security, disable TA transmission.
   // NOTE 2:   For AM DRBs the PDCP count is maintained while for SRBs and UM DRBs the PDCP count is reset to 0.
   if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s191200 R5s200309 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_NR_CellId, p_NCC, cs_TimingInfo_NR(v_Timing), -, v_NR_PdcpCountInfoList); //@sic R5s200754 R5s220123 sic@
   } else {
      v_RRC_SecurityParams := f_NR_Security_Get();
      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false,-, v_NR_PdcpCountInfoList); //@sic R5s220123 sic@
      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering,
                                            cs_TimingInfo_NR(v_Timing)); // @sic R5s191200_r2 sic@
   }
  
   // Disable TA transmission
   f_NR_ULGrantConfiguration_Common(p_NR_CellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, cs_UL_GrantConfig_OnSR);

  }
[bookmark: _Toc92457119]1.2 Correction to f_NR_SS_AS_ActivateSecurity_Current
	Function name
	f_NR_SS_AS_ActivateSecurity_Current()

	Reason for change
	In the function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8, at Step 7, PDCP counts are being set based on an activation time. However, at Step 8, whilst re-establishing security, PDCP counts are being obtained with immediate activation time which would not include the increment of the DL PDCP SN for the transmission of the RRCReconfiguration message at Step 2. Hence, this mechanism does not guarantee the values that are obtained at Step 8 are the ones that were set at Step 7.

	Summary of change
	It is recommended to obtain the PDCP counts independent of the flag p_SetGetOfPDCPCount and increment the DL PDCP SN for the transmission of the RRCReconfiguration message at Step 2 and in the case of RBConfig_KeyChange being TRUE, reset the PDCP count to 0 as well as use the obtained PDCP counts at Step 1 to pass as a parameter to functions that re-establish security.

	TTCN module
	NR_SecuritySteps

	MCC160 Comment
	Accepted



Before Change:
	function f_NR_SS_AS_ActivateSecurity_Current(NR_CellId_Type p_NR_CellId,
                                               NextHopChainingCount p_NCC := tsc_NR_38508_NextHopChainingCount,
                                               template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                               boolean p_IncrSRB1_DL  := false
                                               ) runs on NR_BASE_PTC
  {
    //@sic R5s201221: added p_IncrSRB1_DL. sic@
    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();
    
    // refresh security keys
    v_RRC_SecurityParams := f_NR_InitAS_KeyChaining_KeyRefresh(f_NR_CellInfo_GetPhysicalCellId(p_NR_CellId),
                                                               f_NR_CellInfo_GetFrequencySSB(p_NR_CellId),
                                                               v_RRC_SecurityParams,
                                                               p_NCC);
                                       
    // reset the PDCP sequence numbers for ciphering
    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false, p_IncrSRB1_DL);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                          v_RRC_SecurityParams.AS_Integrity,
                                          v_RRC_SecurityParams.AS_Ciphering,
                                          p_TimingInfo);
    f_NR_Security_Set(v_RRC_SecurityParams);
  }



After Change:
	function f_NR_SS_AS_ActivateSecurity_Current(NR_CellId_Type p_NR_CellId,
                                               NextHopChainingCount p_NCC := tsc_NR_38508_NextHopChainingCount,
                                               template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,
                                               boolean p_IncrSRB1_DL  := false,
                                               template(omit) NR_PdcpCountInfoList_Type    p_PdcpCountInfoList:= omit //WA#
                                               ) runs on NR_BASE_PTC
  {
    //@sic R5s201221: added p_IncrSRB1_DL. sic@
    var NR_SecurityParams_Type v_RRC_SecurityParams := f_NR_Security_Get();
    
    // refresh security keys
    v_RRC_SecurityParams := f_NR_InitAS_KeyChaining_KeyRefresh(f_NR_CellInfo_GetPhysicalCellId(p_NR_CellId),
                                                               f_NR_CellInfo_GetFrequencySSB(p_NR_CellId),
                                                               v_RRC_SecurityParams,
                                                               p_NCC);
                                       
    // reset the PDCP sequence numbers for ciphering
    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_NR_CellId, false, p_IncrSRB1_DL, p_PdcpCountInfoList); //WA#
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_NR_CellId,
                                          v_RRC_SecurityParams.AS_Integrity,
                                          v_RRC_SecurityParams.AS_Ciphering,
                                          p_TimingInfo);
    f_NR_Security_Set(v_RRC_SecurityParams);
  }




1.3 [bookmark: _Toc92457120]Correction to f_NR_RRC_CipherActTime_GetCurrent
	Function name
	f_NR_RRC_CipherActTime_GetCurrent()

	Reason for change
	In the function f_NR5GC_508RRC_IntraNR_HO_IntraCell_Step1_8, at Step 7, PDCP counts are being set based on an activation time. However, at Step 8, whilst re-establishing security, PDCP counts are being obtained with immediate activation time which would not include the increment of the DL PDCP SN for the transmission of the RRCReconfiguration message at Step 2. Hence, this mechanism does not guarantee the values that are obtained at Step 8 are the ones that were set at Step 7.

	Summary of change
	It is recommended to obtain the PDCP counts independent of the flag p_SetGetOfPDCPCount and increment the DL PDCP SN for the transmission of the RRCReconfiguration message at Step 2 and in the case of RBConfig_KeyChange being TRUE, reset the PDCP count to 0 as well as use the obtained PDCP counts at Step 1 to pass as a parameter to functions that re-establish security.

	MCC160 Comment
	Accepted



Before Change:
	function f_NR_RRC_CipherActTime_GetCurrent(NR_CellId_Type p_NR_CellId, boolean p_IncrSRB1, boolean p_IncrSRB1_DL := false ) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type
  {

     <<SKIPPED CODE>>

     v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_NR_CellId);

     <<SKIPPED CODE>>

     



After Change:
	function f_NR_RRC_CipherActTime_GetCurrent(NR_CellId_Type p_NR_CellId, boolean p_IncrSRB1, boolean p_IncrSRB1_DL := false, template(omit) NR_PdcpCountInfoList_Type p_PdcpCountInfoList:= omit) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type
  {

     <<SKIPPED CODE>>

    //WA#
    if (isvalue(p_PdcpCountInfoList)) {
        v_PdcpCountInfoList := valueof(p_PdcpCountInfoList);
    } else {   
       v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_NR_CellId);
    }

    <<SKIPPED CODE>>




